(3GPP SA4-EVS SWG Ad-hoc Meeting #7)/TSG-SA4#70 meeting
Tdoc S4-121007
(11-12 Aug) /13th – 17th Aug, 2012, Chicago, USA


Source:
Fraunhofer IIS

Title:
Cutting tool for EVS JBM simulations
Document for:
Discussion & Approval
Agenda Item:
8
Overview

The source already proposed a trace format for JBM simulations [1] and envisioned that a corresponding tool to allow cutting could be provided. This contribution contains the source’s implementation of a tool that parses the trace files and is then able to convert segmentation command lines (see section 5.1.7 of EVS-7a v0.0.8 [2]) to command lines that reflect the variable delay of the candidates’ JBM solutions. 
The tool is written in Perl. A binary for Win32 is attached for convenience.

Proposed section for EVS-7a
Note that the new sections are provided for 16kHz only for simplicity. Sections for 8 and 32kHz sampling rate are to be formatted accordingly.
CuT Codec (16-kHz sampling) (JBM case)
[

Figure 1: Processing by the CuT with jitter.]
JBM trace files

The CuT codec shall generate a trace file as a by-product of the decoder operation in case of JBM operation (triggered with the –voip switch on the decoder side). The by-product shall be named as Output16.jbmtrace.csv , i.e a suffix .jbmtrace.csv  shall be appended in case of JBM operation.
The trace file is specified as a CSV file with semi-colon as separator. For each played out speech frame an entry shall be written to the trace file. The interval of the playtime values is normally 20ms, but may differ, depending on the CuT operation. Each line of the trace file shall consist of the following columns:

rtpSeqNo;rtpTs;rcvTime;playtime;active
The values are defined as follows:

	
	Unit
	Description

	rtpSeqNo
	1
	RTP sequence number of played out speech frame. -1 if no corresponding RTP packet for the speech frame exists.

	rtpTs
	ms
	RTP time stamp of played out speech frame. -1 if no corresponding RTP packet for the speech frame exists

	rcvTime
	ms
	Absolute reception time of the RTP packet that corresponds to the speech frame. -1 if no corresponding RTP packet for the speech frame exists.

	playtime
	ms
	Absolute play time (i.e. the time at which the PCM data is made available by the decoder). Can be floating-point value.

	Active
	0 or 1
	Binary entry, which is set to 1 for active speech frames (i.e. frames that are neither SID nor NO_DATA)


Conversion of astrip trim marks for JBM-processed files
[

[image: image2.emf]astrip 

segmentation 

values

Trim mark 

conversion

Modified 

astrip 

segmentation 

values

JBM trace file


Figure 1: Conversion of astrip trim marks.]
Windowing and segmentation
In case of JBM for the CuTs, the values s and m below are to be converted by the JBM trim mark conversion tool. Use the following command line to generate a pair of values s’ and m’ for the sampling frequency f from the JBM trace file t:

(s’, m’) = jbmtrim f s m t
The output of the script will be in the form “s’ m’” ,where the output values are separated by a single space character. Those values s’ and m’ shall then be used for the astrip commands below instead of s and m.
Note: The implementation provided outputs a string in the format “–start s’ -n m’” directly for usage in processing scripts.
Note: the following section is copied from [2] and just provided for information, to understand how the tool fits into the current processing plan:

Segmentation for WB experiments

[To extract an m sample long file beginning at sample s from a 16 kHz single channel concatenated file, use:

astrip –sample –smooth –wlen 1600 –start s -n m input16 output16]
Summary
The source provides the cutting tool for the CuTs with JBM in order to fill missing sections in the processing plan. The source code of the compiled binary of jbmtrim is attached for convenience. The binary has been compiled using pp available in ActiveState Perl 5.14.2.
References:

[1] AHEVS-159 – Processing functions for Jitter Buffer Management

[2] EVS-7a – Processing Test plan v0.0.8
Appendix A – jbmtrim.pl:

#!/usr/bin/perl -w
#--------------------------------------------------------------------------#
# Item extraction tool for JBM tests, V1.0                                 #
# ------------------------------------------                               #
# (C) 2012 Fraunhofer IIS. All rights reserved. Provided by Fraunhofer IIS #
# for exclusive use of the extraction of items from the output files of    #
# CuT codecs for JBM tests in the 3GPP SA WG4 EVS codec standardization.   #
# Any other use is not permitted.                                          #
#                                                                          #
# Fraunhofer IIS makes no representation nor warranty in regard to         #
# the accuracy, completeness or sufficiency of The Software, nor           #
# shall Fraunhofer IIS be held liable for any damages whatsoever           #
# relating to use of said Software.                                        #
#--------------------------------------------------------------------------#
use strict;
use warnings;
###############################################################################
############################  Configuration  ##################################
###############################################################################
# hard coded settings
my $rtpTimeScale  = 1000; # RTP time scale of the JBM trace
my $playTimeScale = 1000; # playout time scale of the JBM trace
###############################################################################
############################  LOCAL FUNCTIONS  ################################
###############################################################################
sub mapItemPositionJBM($$$) {
  my $jbmTrace   = shift;
  my $itemStart  = shift;
  my $itemLength = shift;
  # use JBM trace file for extracting the items
  open(jbmTraceFile, $jbmTrace) || die "failed to open JBM trace file: $jbmTrace\n";
  my $lastRtpTs = -1;
  my $lastPlayTime = -1;
  my $currBeginPlayTime = -1;
  my $nextCutTime = $itemStart;
  # read line by line
  while(<jbmTraceFile>) {
    # discard comments (lines starting with '#')
    next if /^#/;
    # split $line into tokens separated by ';'
    my @cells = split(";", $_);
    my $rtpTs = $cells[1];
    my $playTime = $cells[3];
    # ignore frames with unknown RTP timestamp or playout time
    next if $rtpTs == -1;
    next if $playTime == -1;
    # check if the next position to cut is found
    if( $rtpTs / $rtpTimeScale >= $nextCutTime ) {
      # interpolate between current and previous RTP timestamp to
      # increase accuracy in case of DTX where lot of timestamps are missing
      my $rtpTsRelErr = ( $rtpTs / $rtpTimeScale - $nextCutTime ) / ( ( $rtpTs - $lastRtpTs ) / $rtpTimeScale );
      my $playTimeAbsErr = $rtpTsRelErr * ( $playTime - $lastPlayTime ) / $playTimeScale;
      if( $currBeginPlayTime == -1 ) {
        # found begin of item
        $currBeginPlayTime = $playTime / $playTimeScale - $playTimeAbsErr;
        # now look for end of item
        $nextCutTime += $itemLength;
      }
      else {
        # found complete item
        my $currEndPlayTime = $playTime / $playTimeScale - $playTimeAbsErr;
        my $currLengthPlayTime = $currEndPlayTime - $currBeginPlayTime;
        return ( $currBeginPlayTime, $currLengthPlayTime );
      }
    }
    $lastRtpTs = $rtpTs;
    $lastPlayTime = $playTime;
  }
  # check if item begin was found
  if( $currBeginPlayTime == -1 ) {
    die "Invalid item start position specified!\n";
  }
  # return item with missing end
  my $currEndPlayTime = $lastPlayTime / $playTimeScale;
  my $currLengthPlayTime = $currEndPlayTime - $currBeginPlayTime;
  return ( $currBeginPlayTime, $currLengthPlayTime );
}
###############################################################################
############################  MAIN  ###########################################
###############################################################################
# Print help if no arguments
if( scalar(@ARGV) != 4 ) {
  printf ("Script to extract items processed by JBM: Maps original item position to astrip parameters for JBM tests.\n");
  printf ("Usage:\n");
  printf ("  %s f s m t\n",$0);
  printf ("where \n");
  printf ("  f = sampling frequency of audio file.\n");
  printf ("  s = original item start position [samples].\n");
  printf ("  m = original item length [samples].\n");
  printf ("  t = JBM trace file name (created by decoder).\n");
  printf ("Output:\n");
  printf ("  -start s' -n m'\n");
  printf ("where\n");
  printf ("  s' = item start position after JBM processing [samples].\n");
  printf ("  m' = item length after JBM processing [samples].\n");
  printf ("\n");
  exit;
}
# Init arguments
my $f = shift;
my $s = shift;
my $m = shift;
my $t = shift;
# Sanity checks
$s > 0  || die "Invalid sampling frequency specified!\n";
$s >= 0 || die "Invalid item start position specified!\n";
$m > 0  || die "Invalid item length specified!\n";
my ($s2, $m2) = mapItemPositionJBM( $t, $s / $f, $m / $f );
# print processed item position as parameter for the astrip tool
printf( "-start %f -n %f\n", $s2 * $f, $m2 * $f );

